"Musha hierba" by Hython

Include Spanish by Sebastian Arg.
Include Decorados by El Enano Malauva.
Include Basic Screen Effects Sp by Emily Short.

Part 1 - Reglas generales modificadas

Chapter 1 - Utilidades o librerías integradas

Section 1 - Facilidad para mostrar partes no interactivas

[--------------------------------------------------

La idea es bien simple -ya que hay muchas cosas 
que no sé hacer-, cada entrada de una tabla/lista 
simple, se muestra y tras ella se espera la 
pulsación de una tecla

--------------------------------------------------]
To relatar fragmento en (tabla - a table-name):
	let N be the number of rows in tabla;
	repeat through tabla begin;
		decrease N by 1;
		say "[frase entry]";
		say paragraph break;
		if N is not 0, esperar pulsacion de tecla;
	end repeat.

To relatar fragmento en (tabla - a table-name), limpiando pantalla:
	relatar fragmento en tabla;
	esperar pulsacion de tecla;
	clear the screen.


Part  2 - Musha hierba

Chapter 0 - El comienzo

[--------------------------------------------------

   Regla activada al iniciar el juego

--------------------------------------------------]
When play begins:
	poner cuadro de inicio;
	wait for any key;
   	clear the screen;
	contar inicio de la historia;
	now the description of the player is "Ni sé quién soy, ni importa... ¡con la de puzzles que hay que resolver y tu preocupándote por tí mismo!.";
	poner cuadro de inicio real;
	wait for any key;
	clear the screen.

[FRASE: Cuadro de inicio con el nombre del relato y fecha]
To poner cuadro de inicio:
	display the boxed quotation
	"Perdida en la hierba



	Un relato de Mel Hython.
	Año 2008.";
	show the current quotation.

To poner cuadro de inicio real:
	display the boxed quotation
	"Musha hierba



	Un relato de Hython sin Mel.
	Año 200x.";
	show the current quotation.


[FRASE: Introducción que relata el despertar de Layna]
To contar inicio de la historia:
	relatar fragmento en Table of Historia Fallida, limpiando pantalla.

[HISTORIA: El fragmento inicial]
Table of Historia Fallida
frase
"Esto... a estas alturas debería estar completada la segunda parte de las aventuras de Layna."
"[italic type]Ya te digo que debería.[roman type]"
"...pero ha habido una gran cantidad de problemas técnicos inexplicables..."
"[italic type]Básicamente vagancia.[roman type]"
"...de forma que nos resulta imposible mostrarles una aventura de Mel Hython, adecuadamente terminada..."
"[italic type]¿Pero tiene alguna?[roman type]" 
"...sin embargo, ya que sabemos el esfuerzo que representa arrancar una aventura hoy en dia..."
"[italic type](al menos hasta que funcione completamente puertAventura)[roman type]"
"...vamos a ofrecerle una aventura con los elementos que encontremos disponibles, respetando en lo posible las reglas no escritas de este género."
"[italic type]Oh, oh, esto suena mal[roman type]"
"Esperemos que la disfrute."

Chapter 1 - El Lugar

Entre la hierba is a habitacionDecorada. The description of entre la hierba is "Hierba. Musha hierba por toas partes.". The decoracion of entre la hierba is Table of DecoHierba.

Table of DecoHierba
topic			descripcion							printedName
"hierba" or "hierbas"	"Qué va a ser, pues hierba, ya sabes."	"las hierbas"
"pie" or "pies"		"Son pies. Tus pies. De hecho incluso podrías asegurar que no hay otros pies que sean tus pies."	"los pinreles"


Chapter 2 - Acciones

[Todo excepto lo que vayamos poniendo]
Instead of doing anything except going or examining or examining decoration or digging or looking or switching on or switching off or smoking or smokking or smokting or putting on or taking or taking inventory:
	say "No, Planseldom, no es necesario hacer eso. Prueba otra cosa, algo relacionado con la hierba probablemente.".

Instead of taking mechero:
	say "Sí, esto sí que hay que cogerlo.";
	move mechero to player.

Instead of taking arenque rojo:
	say "¡Por favor! ¡Coger un arenque rojo?? ¿Y las reglas no escritas para qué están?".

Instead of taking anything:
	say "Eso no hay que cogerlo para nada, Planseldom, piensa en otra cosa.".

[Moverse]
Instead of going nowhere:
	say "Caminas unos pasos entre la hierba y solo encuentras más hierba. O sea, que no hay a donde ir.".

[Mirar arriba o abajo]
Instead of examining down:
	say "A ver... tus pies que son.. normales y tierra... y... talllos de las hierbas, y supongo que si excavas encontrará raices de las hierbas. Vamos, toda una sorpresa.".

Instead of examining a direction:
	say "Miras hacia [the noun], y no encuentras nada especial. Ni detalles extra ni nada. Parece que aquí solo hay hierba por todas partes.".

[Cavar]
Digging is an action applying to nothing. Understand "cavar" or "cava" as digging.

Carry out digging:
	say "Excavas. Tierra. Claro. ¿Qué si no? Un momento, ¡un mechero!";
	move mechero to entre la hierba.

[Fumar]
Smoking is an action applying to nothing. Understand "fuma la hierba" or "fumar la hierba" or "fumate la hierba" or "fumarse la hierba" or "fuma hierba" or "fumar hierba" or "fumarse hierba" or "fumate hierba" as smoking.

Carry out smoking:
	if mechero is carried:
		if mechero is switched off:
			say "Imposible con el mechero apagado.";
		otherwise:
			say "Agarras toda la hierba que hay a tu al rededor, y le das el uso que se merece.";
			wait for any key;
			say "Oh... probablemente no haya nada tan estupendo, excepto tal vez resolver un puzzle, uno bueno... ya sabes... algo con un tablero y muchas piezas.";
			end the game in victory;
	otherwise:
		say "Sin llevar encima un mechero imposible.".
		


Smokking is an action applying to nothing. Understand "fuma" or "fumar" as smokking.
Smokting is an action applying to one thing. Understand "fuma [a thing]" or "fumar [a thing]" or "fumate [a thing]" or "fumarse [a thing]" as smokting.

Carry out smokking:
	say "Pero, ¿qué hay que fumar?".

Carry out smokting:
	say "No, no, Planseldom, eso no es lo que hay que fumarse.".

Instead of smokting mechero:
	say "Nada, no hay forma de fumarse prender con el mechero al mechero. Será que hay que fumarse otra cosa.".

[Armar el puzzle]
After putting pieza on puzzle:
	say "¡Vaya! Encaja perfectamente, descubriendo un misterioso dibujo: una pala, un cubo y una línea que parece indicar la tierra. ¿Qué significará?".


Chapter 3 - Objetos

[Mechero]
A mechero is a device.

[Puzzle]
A puzzle is in entre la hierba. It is a supporter. The description of the puzzle is "[if pieza is on puzzle]La pieza está en su lugar, ¡por fin![otherwise]Se trata de un puzzle de una pieza. La figura recortada y vacía es desafiante. Debe ser algo muy complicado.[end if]".

[Pieza]
A pieza is carried. It is female. The description of the pieza is "[if pieza is on puzzle]Encaja perfectamente.[otherwise]¡La pieza! ¡Si supieses como encaja! Seguro que te desvelaba un montón de misterios... o el siguiente puzzle a resolver.[end if]".

[Arenque rojo]
A arenque rojo is a thing. It is in entre la hierba. The description of arenque rojo is "Un pez muerto en mitad de la nada. Se podría pensar que seguro que sirve para algo, pero es un arenque rojo."

Chapter 4 - Detallines

Rule for printing a parser error:
	say "Planseldom, vas muy desencaminado, esto ni lo entiende el parser, tío." instead.

